Includes easy-entry keystroke listings.

HEWLETT-PACKARD

HP-75

USERS' LIBRARY SOLUTIONS

Statistics

NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

HEWLETT PACKARD

HP-75 SOLUTIONS BOOK/MEDIA ORDER FORM

All HP-75 Solutions Books are available recorded from the Users' Library* or from any Hewlett-Packard Full Service Dealer. The individual Solutions Books are $12.50 each; recorded mini-data cassettes are $12.00 for each Solutions Book requested.

Telephone the Users' Library directly (503-757-2000, ext. 3371) with your Solutions Book request, your order will ship that SAME DAY; or, call our TOLL FREE number (800-547-3400, Continental U.S. only), and orders ship within 48 hours. Mail requests also ship within 48 hours of receipt.

HP-75 Users' Library Solutions Books

00075-90061	Math I
00075-90062	Math II
00075-90063	Math III
00075-90065	Games
00075-90066	Electronics
00075-90067	Statistics
00075-90070	Real Estate
00075-90071	Finance
00075-90076	Test Statistics

Solutions Book Part Number	Title	Book Only	Mini-Data Cassettes	Price
00075-90_____				
00075-90_____				
00075-90_____				
	Please add Handling Charge on orders including Solutions Books			**3.50**
			State and Local Taxes	
			TOTAL	

All payments must be in U.S. dollars and drawn on a U.S. bank.
Send orders (with payment) directly to: HEWLETT-PACKARD USERS' LIBRARY
1000 N.E. Circle Blvd.
Corvallis, Oregon 97330

☐ Payment enclosed ☐ Master Card ☐ VISA ☐ Purchase order

Card Number/Company Purchase Order Exp

Name

Company (if applicable)

Address Telephone

City State/Country Zip

*Recorded Solutions Book programs shipped to U.S. customers only.

TABLE OF CONTENTS

PROGRAM DESCRIPTION

BASICS - BASIC ONE VARIABLE STATISTICS

This program computes the basic statistics of grouped or ungrouped data. The functions computed are: Mean, variance (with population and sample), standard deviation (with population and sample), standard error, minimum value, and maximum value.

Formulae:

$$\text{Mean } (x) = \frac{\Sigma x}{n}$$

$$\text{Sample variance} = \frac{\Sigma x^2}{n} - \left(\frac{\Sigma x}{n}\right)^2$$

$$\text{Sample standard deviation } s = \sqrt{\text{Sample variance}}$$

$$\text{Population estimate of variance} = (\text{Sample variance}) \frac{n}{n-1}$$

$$\text{Standard error} = \frac{\text{Sample S.D.}}{\sqrt{n-1}}$$

SAMPLE PROBLEM

Compute the basic statistics of the following grouped and ungrouped data:

Grouped data				
Value	5	3	6	2
Frequency	1	4	7	11

Ungrouped data							
1	5	4	2	6	7	4	7

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run "BASICS"	Basic One - Variable Statistics	
2	Select grouped data	Grouped or Ungrouped data?	G [RTN]
3		RTN to stop D,D to correct	
4	Start data input	Enter value 1 , freq. ?	5,1 [RTN]
		Enter value 2 , freq. ?	3,4 [RTN]
	Error	Enter value 3 , freq. ?	76,77 [RTN]
6	Delete error	Enter value 4 , freq. ?	D,D [RTN]
		Value 3 deleted	
4	Enter correct value	Enter value 3 , freq. ?	6,7 [RTN]
		Enter value 4 , freq. ?	2,11 [RTN]
7	Stop data input	Enter value 5 , freq. ?	[RTN]
8	View statistics:	N of values = 23	[RTN]
	Mean	Mean = 3.52173	[RTN]
	True population std. deviation	s (population) = 1.76609	[RTN]
	Sample estimate std. deviation	s (sample) = 1.80578	[RTN]
	True population variance	Variance (pop) = 3.11909	[RTN]
	Sample estimate variance	Variance (sample) = 3.26086	[RTN]
	Std. error of the mean	St. Err. Mean = .37653	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Minimum value	Minimum value = 2	[RTN]
	Maximum value	Maximum value = 6	[RTN]
9	Run again	Run, View, Continue or End?	R [RTN]
2	Select data type	Grouped or Ungrouped data?	U [RTN]
3		RTN to stop, D to correct	
5	Enter data values	Enter value 1 ?	1 [RTN]
		Enter value 2 ?	5 [RTN]
		Enter value 3 ?	4 [RTN]
		Enter value 4 ?	2 [RTN]
		Enter value 5 ?	6 [RTN]
		Enter value 6 ?	7 [RTN]
		Enter value 7 ?	4 [RTN]
		Enter value 8 ?	7 [RTN]
7	Stop data input	Enter value 9 ?	[RTN]
8	View statistics	N of values = 8	[RTN]
	Mean	Mean = 4.5	[RTN]
	True population std. deviation	s (population) = 2.06155	[RTN]
	Sample estimate std. deviation	s (sample) = 2.20389	[RTN]
	True population variance	Variance (pop) = 4.25	[RTN]
	Sample estimate variance	Variance (sample) = 4.85714	[RTN]
	Std. error of the mean	St. Err. Mean = .77919	[RTN]
	Minimum value	Minimum value = 1	[RTN]
	Maximum value	Maximum value = 7	[RTN]
9	End program	Run, View, Continue or End?	E [RTN]

If 'R' then re-run program
If 'C' then continue to add data to present data set
If 'V' re-view results
If 'E' then exit program

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run "BASICS"	Basic One-Variable Statistics	
2	Select type of input	Grouped or Ungrouped data?	G or U [RTN]
3	For grouped data goto step 6		
4	Enter upgrouped data:	RTN to stop, D to correct	
5	Enter value	Enter value j	X_j [RTN]
	To delete last entry	Enter value j	D [RTN]
	Goto step 5	Value j-1 (X_{j-1}) deleted	
	To end data entry press [RTN]		
	and goto step 8	Enter value j	[RTN]
6	Enter grouped data	RTN to stop, D,D to correct	
7	Enter value and frequency	Enter value j , Freq. ?	X_j ,F_j [RTN]
	To delete last entry	Enter value j , Freq. ?	D,D [RTN]
	Goto step 7	Value j-1 (X_{j-1}*F_{j-1}) deleted	
	To end data entry press [RTN]	Enter value j , Freq. ?	[RTN]
8	View statistics:	N of values =	[RTN]
		s (population) =	[RTN]/[BACK]
		s (sample) =	[RTN]/[BACK]
		Variance (pop) -	[RTN]/[BACK]
		Variance (sample) -	[RTN]/[BACK]
		St. Err. Mean =	[RTN]/[BACK]
		Minimum value =	[RTN]/[BACK]
		Maximum value =	[RTN]/[BACK]
9	Continuation options	Run, View, Continue, or End?	R, V, C or E [RTN]
	If 'R' then run program again, goto step 2 If 'V' then view statistics again, goto step 8		
	If 'C' then add more data, goto step 3 If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A( )	Data storage	N$	User input string
F( )	Frequency storage	A$	Data input string
N	Number of data points	B$	Frequency input string
N5	Sample size	X2	Sum of X squared
T	Sum of X	G	Frequency value
G9	Flag: 1=grouped data	M	Mean of data
V1	Variance (true population)	S9	Standard deviation (pop)
V2	Variance (sample estimate)	S8	Standard dev. (sample)
E3	Standard error of mean	I	Minimum value
A	Maximum value		

NOTES AND REFERENCES

The maximum number of data points (a grouped value is one data point) that may be entered is 300. To change this, change the dimension statements in line 60.

PROGRAM LISTING

```
 10 ! BASIC5 - basic one
 20 ! variable stat.
 30 ! REV 11/01/82
 40 !
 50 DISP "Basic One-Variable Statistics
    " @ DELAY .5
 60 DIM A(300),N$[25],F(300),A$[25],B$[
    25]
 70 DEF FNI(X) = INT(X*100000)/100000      -Routine to set output
                                            precision
 80 DEF FNE
 90 IF N<1 THEN BEEP @ DISP 'Must have     -Routine to delete last entry
    data to delete' @ FNE=1 @ GOTO 170
100 DISP "Value";N;" (";A(N); @ IF F(N)
    >1 THEN DISP '*';F(N);
110 DISP ') deleted' @ BEEP 440 @ WAIT
    1
120 N5=N5-F(N) @ X2=X2-A(N)^2*F(N) @ T=   -Correct counters
    T-A(N)*F(N) @ N=N-1 @ FNE=0
130 I=INF @ A=EPS @ FOR Q=1 TO N
140 IF A(Q)>A THEN A=A(Q)
150 IF A(Q)<I THEN I=A(Q)
160 NEXT Q
170 END DEF
180 DEF FND                               -Wait for RTN or BACK keys.
                                            Return 1 if BACK key is used
190 BEEP 440
200 K$=KEY$ @ IF K$="" THEN 200
210 FND=K$=CHR$(8) @ IF K$<>CHR$(8) AND
     K$<>CHR$(13) THEN 200
220 END DEF
230 !
240 ! initialize
250 !
260 DISP CHR$(199);"rouped or ";CHR$(21   -Start up dialogue
    3);"ngrouped data "; @ INPUT N$ @ N
    $=N$&" "
270 G9=UPRC$(N$[1,1])="G"
280 IF NOT G9 THEN DISP 'RTN to stop, D
     to correct' @ WAIT 1
290 IF G9 THEN DISP 'RTN to stop, D,D t
    o correct' @ WAIT 1
300 ! Clear counters
310 !
320 T,X2,N,N5=0 @ I=INF @ A=EPS
330 !
340 ! Enter data
350 !
360 ON ERROR GOTO 430
370 DISP "Enter value";N+1;               -Data entry loop
380 G=1 @ IF G9 THEN 410
390 INPUT A$ @ IF A$="" THEN 540
400 B$="" @ GOTO 460
410 DISP ", freq."; @ INPUT A$,B$
```

PROGRAM LISTING

```
420 GOTO 460
430 IF G9 AND ERRN=81 THEN A$,B$="" @ G
    OTO 540
440 BEEP 1680 @ DISP "Illegal data - pl
    ease retype" @ GOTO 370
450 INPUT A$,B$ @ IF A$="" THEN 540
460 IF POS('EeDd',A$) THEN V=FNE @ GOTO
     370                              -Call error correction routine
470 IF B$<>"" AND G9=1 THEN G=ABS(INT(V
    AL(B$)))                         -Exit input mode
480 X=VAL(A$)
490 N=N+1 @ N5=N5+G @ A(N)=X @ F(N)=G    -Increment counters
500 T=T+X*G @ X2=X2+X^2*G
510 IF X<I THEN I=X
520 IF X>A THEN A=X
530 IF N<301 THEN 370
540 IF N<2 THEN BEEP @ DISP 'Must have
    more than one subject' @ GOTO 370
550 OFF ERROR
560 !
570 ! Compute Mean, s
580 !
590 V1=X2/N5-(T/N5)^2 @ S9=SQR(V1)       -Compute variance and standard
                                          deviation for N
600 V2=V1*(N5/(N5-1)) @ S8=SQR(V2)       -Compute variance and standard
                                          deviation for N-1
610 M=T/N5 @ E3=S9/SQR(N5-1)             -Compute mean and standard
                                          error
620 !
630 ! Print-out
640 !
650 PRINT "N of values =";N5 @ U=FND     -Output results and call delay
                                          routine
660 PRINT "Mean =";FNI(M) @ U=FND @ IF
    U THEN 650
670 PRINT "s (population) =";FNI(S9) @
    U=FND @ IF U THEN 660
680 PRINT "s (sample) =";FNI(S8) @ U=FN
    D @ IF U THEN 670
690 PRINT "Variance (pop) =";FNI(V1) @
    U=FND @ IF U THEN 680
700 PRINT "Variance (sample) =";FNI(V2)
     @ U=FND @ IF U THEN 690
710 PRINT "St. Err. Mean =";FNI(E3) @ U
    =FND @ IF U THEN 700
720 PRINT "Minimum value =";I @ U=FND @
     IF U THEN 710
730 PRINT "Maximum value =";A @ U=FND @
     IF U THEN 720
740 !
750 ! End dialogue
760 !
770 DISP CHR$(210);"un, ";CHR$(214);"ie    -Continuation options
    w, ";CHR$(195);"ontinue, or ";CHR$(
    197);"nd";
```

PROGRAM LISTING

```
780 INPUT N$ @ N$=N$&" " @ N$=UPRC$(N$[
    1,1])
790 ON POS("RVCE",N$)+1 GOTO 770,260,65
    0,370,800
800 STOP
```

PROGRAM DESCRIPTION

CORR - COEFFICIENT OF CORRELATION

This program calculates a Pearson Product-moment correlation matrix and the basic statistics of a N*K sized data matrix.

Formulae:

$$\text{Mean of column} = \frac{\sum_{i=1}^{n} X_i}{n}$$

$$\text{Standard deviation} = \left[\frac{\sum_{i=1}^{n} X_i^2}{n} - \left(\frac{\sum_{i=1}^{n} X_i}{n} \right)^2 \right]^{1/2}$$

$$\text{Correlation } r = \frac{n\sum_{i=1}^{n} X_i Y_i - \sum_{i=1}^{n} X_i \sum_{i=1}^{n} Y_i}{\left\{ \left[n\sum_{i=1}^{n} X_i^2 - \left(\sum_{i=1}^{n} X_i \right)^2 \right] \left[n\sum_{i=1}^{n} Y_i^2 - \left(\sum_{i=1}^{n} Y_i \right)^2 \right] \right\}^{1/2}}$$

Where n = the number of subjects

SAMPLE PROBLEM

The following are the scores of six subjects on 3 psychological tests. Determine the correlations among the tests and the mean and standard deviation (N) of each test:

Subject	Test 1	Test 2	Test 3
1	82	16	18
2	14	12	29
3	05	14	56
4	09	18	11
5	18	11	17
6	11	17	41

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "CORR"	Coefficient of Correlation	
1	Enter the N of subjects	How many subjects?	6 [RTN]
2	Enter the N of variables	How many variables?	3 [RTN]
3	Enter data	Subject 1 , Item 1 ?	82 [RTN]
		Subject 1 , Item 2 ?	16 [RTN]
		Subject 1 , Item 3 ?	18 [RTN]
	Subject 2	Subject 2 , Item 1 ?	14 [RTN]
		Subject 2 , Item 2 ?	12 [RTN]
		Subject 2 , Item 3 ?	29 [RTN]
	Subject 3	Subject 3 , Item 1 ?	5 [RTN]
		Subject 3 , Item 2 ?	14 [RTN]
		Subject 3 , Item 3 ?	56 [RTN]
	Subject 4	Subject 4 , Item 1 ?	9 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Subject 4 , Item 2 ?	18 [RTN]
	Error	Subject 4 , Item 3 ?	110 [RTN]
	Subject 5	Subject 5 , Item 1 ?	18 [RTN]
		Subject 5 , Item 2 ?	11 [RTN]
		Subject 5 , Item 3 ?	17 [RTN]
	Subject 6	Subject 6 , Item 1 ?	11 [RTN]
		Subject 6 , Item 2 ?	17 [RTN]
		Subject 6 , Item 3 ?	41 [RTN]
5	Correct error	OK? (N=change, RTN=continue)?	N [RTN]
6	Enter index	Change subject # , Item # ?	4,3 [RTN]
7	Overstrike	Subject 4 , Item 3 Value : 110	11[DEL][RTN]
5	Continue	OK? (N=change, RTN=continue)?	[RTN]
9	Read display	Clm 1 Me=23.1667 s=26.6171	[RTN]
		Corr (1:2) = .1427	[RTN]
		Corr (1:3) = -.3879	[RTN]
		Clm 2 Me=14.6667 s=2.5604	[RTN]
		Corr (2:3) = -.0862	[RTN]
		Clm 3 Me=28.6667 s=15.6063	[RTN]
10	Recycle prompt	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "CORR"	Coefficient of Correlation	
1	Enter the number of subjects	How many subjects?	r [RTN]
2	Enter the number of variables	How many variables?	c [RTN]
3	Enter data as prompted, goto		
	step 3 until finished	Subject i variable j?	A_{ij} [RTN]
4	Indicate error:	OK? (N=change, RTN=continue)?	N [RTN] or [RTN]
	If no error, goto step 7		
5	Enter index of bad data item	Change subject # , Item # ?	E_r, E_c [RTN]
6	Correct value and goto step 4	$Subject_r$ $Item_c$ $Value_{ac}$	value [RTN]
7	Read display	Clm_c Me_m s_s	[RTN] or [BACK]
8	Program options	Run again, View again, or End?	R, V or E [RTN]
	If 'R' then re-run program		
	If 'V' then review (step 7)		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A(,)	Data matrix	T	Sum of all data Σx
T(j)	Sum of clm j Σx_j	X2(j)	Sum of clm j Σx_j^2
X2	Sum^2 of all data Σx^2	N$	Correction string
M(j)	Mean of clm. j	S9(j)	s of clm. j
A$	User input string	X,Y & Z	Loop counters
R	N of observations	C	N of variables
CO(,)	Under diagonal=cross product. Over diagonal= correlation coefficient.		

NOTES AND REFERENCES

This program is set up to handle a maximum of 50 subjects and 10 variables. To change this, change lines 50, 190, and 200.

References: Cohen & Cohen, APPLIED MULTIPLE REGRESSION/CORRELATION (Lawrence Erlbaum, New Jersey, 1975) page 35.

PROGRAM LISTING

```
 10 ! CORR - N variable
 20 ! Correlation Coeff.
 30 ! REV 11/01/82
 40 !
 50 DIM A(50,10),T(10),X2(10),M(10),C0(
    10,10),S9(10),A$[10],N$[10]          -Set up array tables
 60 DEF FNI(X) = INT(X*10000+.5)/10000   -Define output precision
 70 X2=0 @ T=0
 80 !
 90 ! Delay Function                      -Delay routine returns 1 for
                                          BACK key, 0 for other keys
100 DEF FND
110 BEEP 440
120 A$=KEY$ @ IF A$<>CHR$(13) AND A$<>C
    HR$(8) THEN 120
130 FND=A$=CHR$(8)
140 END DEF
150 DISP "  Coefficient of Correlation"
160 !
170 ! Input dialogue
180 !
190 INPUT "How many Subjects ?"; R @ IF   -Enter and check data
     R<1 OR R>50 THEN BEEP @ RUN          parameters
200 INPUT "How many Variables ?"; C @ I
    F C<1 OR C>20 THEN BEEP @ RUN
210 FOR Y=1 TO C @ X2(Y)=0 @ T(Y)=0 @ F   -Set all counters to zero
    OR X=1 TO C @ C0(X,Y)=0 @ NEXT X @
    NEXT Y
220 !
230 ! Enter Data
240 !
250 FOR X=1 TO R @ FOR Y=1 TO C           -Matrix input loop
260 DISP "Subject";X;"Item";Y; @ INPUT
    A(X,Y)
270 NEXT Y @ BEEP 440 @ NEXT X
280 !
290 ! Edit Routine
300 !
310 BEEP 220,.1 @ INPUT "Ok? (N=Change,   -Edit routine
     RTN=Continue) ?"; X$ @ X$=X$&" "
320 IF UPRC$(X$[1,1])<>"N" THEN 420       -Branch out if data is correct
330 INPUT "Change Subject#, Item# ?";X,
    Y
340 IF X<1 OR X>R OR Y<1 OR Y>C THEN DI
    SP 'No Subject';X;'Item';Y @ WAIT 3
     @ BEEP @ GOTO 310
350 DISP 'Subject';X;'Item';Y;
360 N$=STR$(A(X,Y)) @ INPUT "Value:",N$
    ;N$
370 A(X,Y)=VAL(N$)
380 GOTO 310
390 !
400 ! Computation
410 !
```

PROGRAM LISTING

```
420 FOR Y=1 TO C                                    -Computation loops
430 FOR X=1 TO R @ IF Y=C THEN 520
440 !
450 ! Cross Product
460 !
470 FOR Z=Y+1 TO C                                  -Compute cross product
480 C0(Z,Y)=C0(Z,Y)+A(X,Y)*A(X,Z) @ NEX
    T Z
490 !
500 ! Me,Std,x,x^2
510 !
520 T(Y)=T(Y)+A(X,Y) @ X2(Y)=X2(Y)+A(X,          -Compute sum X, X squared, sum
    Y)^2 @ T=T+A(X,Y) @ X2=X2+A(X,Y)                 total, sum squared total
530 NEXT X
540 M(Y)=T(Y)/R @ S9(Y)=SQR(X2(Y)/R-(T(             -Compute sample std. deviation
    Y)/R)^2)                                         and mean
550 NEXT Y
560 !
570 ! Compute r
580 !
590 FOR Y=1 TO C                                    -Computation loops
600 IF Y=C THEN 660
610 FOR Z=Y+1 TO C
620 C0(Y,Z)=R*C0(Z,Y)-T(Y)*T(Z)                     -Compute the correlation of Y
                                                     to Z
630 C0(Y,Z)=C0(Y,Z)/(SQR(R*X2(Y)-T(Y)^2
    )*SQR(R*X2(Z)-T(Z)^2))
640 NEXT Z
650 NEXT Y
660 !
670 ! Printout
680 !
690 X,Y,Z=0
700 Y=Y+1 @ IF Y>C THEN 820                         -Printout
710 PRINT "Clm";Y;"Me=";FNI(M(Y));"s=";
    FNI(S9(Y)) @ A=FND
720 IF A=1 AND Y>1 THEN Y=Y-1 @ GOTO 71
    0
730 Z=Y+1 @ IF Y=C THEN 820
740 PRINT "Corr.[";Y;":";Z;"] =";FNI(C0
    (Y,Z)) @ A=FND
750 IF A AND Z>Y+1 THEN Z=Z-1 @ GOTO 74
    0
760 IF A AND Z=Y+1 THEN 710
770 Z=Z+1 @ IF Z<=C THEN 740
780 GOTO 700
790 !
800 ! Recycle
810 !
820 DISP CHR$(210);"un again, ";CHR$(21             -Program options
    4);"iew again, or ";CHR$(197);"nd";
830 INPUT A$ @ A$=UPRC$(A$&" ")
840 ON POS('RVE',A$[1,1])+1 GOTO 820,19
    0,690,850
850 STOP
```

PROGRAM DESCRIPTION

DISTR: PROBABILITY OF NORMAL, F, t, & CHI-SQUARE DISTRIBUTIONS

Normal Distribution:

This program evaluates the right tail of the normal probability distribution. The probability statement (p.) is in the form describing the chances of a score falling above z. The probabilities are accurate to two places when z is less than 2.51.

The cumulative probability of z=1-p.

Chi-Square Distribution:

This program evaluates Chi-Square such that p equals the probability of obtaining a sample chi-square this large or larger. The program is accurate to two places through 70 df.

F. Distribution:

This program evaluates the integral of the F-distribution for given values of F (>0), degrees of freedom V1, V2, whether or not V1 or V2 is even. V1 is the degrees of freedom in the numerator. The p presented is the probability of rejecting the equivalence of variances.

t Distribution:

This program evaluates the cumulative distribution P(x) for a given t and degrees of freedom. The two tailed case is given. P is equivalent to the probability of a false rejection of the null hypothesis.

Caution is recommended when using this program as at most points the p is an approximation of the same order as the condensed table in Biometrics Tables for Statisticians.

General:

This series allows the efficient and relatively accurate generation of probabilities for z, t, f, and Chi-Square. The formulations are based on the work of N. Jaspen and C. Hastings. The approximations are sufficient for approximate evaluations of the specified statistics. More exact evaluations require individual programs or the use of published tables where the accuracies are public.

PROGRAM DESCRIPTION

DISTR: PROBABILITY OF NORMAL, F, t, & CHI-SQUARE DISTRIBUTIONS (continued)

Distribution formulae:

F = F value
I = DF numerator
J = DF denominator

Normalization of F = Z = $$\frac{(1-2/9/J)*F^{1/3}-(1-2/9/I)}{\sqrt{(2/9/J)F^{2/3}+2/9/I}}$$

If J 3 then: $Z' = Z(1+.08*Z^4/J^3)$

Probability $P = .5/(1+C_1*Z+C_2*Z^2+C_3*Z^3+C_4*Z^4)^2$

Where: $C_1 = .196854$, $C_2 = .115194$, $C_3 = .000344$, $C_4 = .019527$

To transform to two-tailed students t, let F = t , I = 1, J = DF

To transform to Chi-squared, let $F = X^2/DF$, I = DF, $J \cong \infty$

Normal probability: z = z score

SAMPLE PROBLEM

Compute the following distributions:

a. Normal Distribution: z=1.18 and z=2.28
b. Chi-square distribution: X2=31.41 V=20
c. F distribution: F=4.21 V1=7 V2=6
d. T distribution: t=2.201 V=11

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "DISTR"	Distribution Package	
1	Select Normal distribution	Normal, Chi^2, F, or T ?	N [RTN]
		Type E at beep to escape	
2.A	Enter z value	Enter z ?	1.18 [RTN]
3	Read probability	Probability = .119	[RTN]
2.A	Enter z value	Enter z ?	2.28 [RTN]
3	Real probability and escape	Probability = .011	E [RTN]
	Run "DISTR"	Distribution Package	
1	Select Chi-square	Normal, Chi^2, F, or T ?	C [RTN]
		Type E at beep to escape	
2.B	Enter Chi-square and DF separ-		
	ated by a comma	Enter Chi-square, DF ?	31.41,20 [RTN]
3	Read probability and escape	Probability = .05	E [RTN]
	Run "DISTR"	Distribution Package	
1	Select F distribution	Normal, Chi^2, F, or T ?	F [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Type E at beep to escape	
2.C	Enter F, DF1, DF2 on the same		
	line, separated by commas	Enter F, DFnum, DFden ?	4.21,7,6 [RTN]
3	Read probability and escape	Probability - .051	E [RTN]
	Run "DISTR"	Distribution Package	
1	Select t distribution	Normal, Chi^2, F, or T ?	T [RTN]
		Type E at beep to escape	
2.D	Enter t value and degrees of		
	freedom	Enter T, DF	2.201,11 [RTN]
3	Read probability and escape	Probability = .048	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "DISTR"	Distribution Package	
1	Select type of distribution	Normal, Chi^2, F, or T	
1.A	N [RTN] = normal		N [RTN]
1.B	C [RTN] = chi-square		C [RTN]
1.C	F [RTN] = F distribution		F [RTN]
1.D	T [RTN] = t distribution		T [RTN]
		Type E at beep to escape	
2	Enter prompted values		
2.A	Normal distribution	Enter z ?	z value [RTN]
2.B	Chi-square	Enter Chi^2, DF ?	X2, DF [RTN]
2.C	F distribution	Enter F, DFnum, DFden ?	F, d1, d2 [RTN]
2.D	t distribution	Enter t, DF ?	t, DF [RTN]
3	Read probability, type [RTN]	Probability = . . .	[RTN] or
	for more of the same type, or		E [RTN]
	type 'E' to exit the program.		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A$	User input string	K$	Delay routine
T	T value	J	Df in denominator
F	F value (or converted t,X^2)	I	Df in numerator
R1	Flag: if F 1 then R1=1	Z	Normalization of F
P	Probability	C	Chi-square value

NOTES AND REFERENCES

The t-distribution is calculated for two-tailed t. To convert to one-tailed t, divide the P-value by 2.

References: Jaspen, Nathan, "The calculation of probabilities corresponding to the values of z, t, F and Chi-square", JOURNAL OF EDUCATIONAL AND PSYCHOLOGICAL MEASUREMENT, page 877-80, Vol XXV, No. 3, 1965.

Fisher & Yates, STATISTICAL TABLES FOR BIOLOGICAL, AGRICULTURAL, AND MEDICAL RESEARCH (6th ED, Oliver and Boyd, London, 1963) page 46-47.

Kerlinger & Pedhauzer, MULTIPLE REGRESSION IN BEHAVORIAL RESEARCH (Holt, Rinehart & Winston, New York, 1973) page 494,499-500.

PROGRAM LISTING

```
 10 ! DISTR - Chi^2, F,
 20 ! & 2-way student's t
 30 ! distributions.
 40 ! REV 11/01/82
 50 !
 60 DISP "    Distribution Package" @ W
    AIT .7
 70 DIM A$[100],K$[100]
 80 DEF FNI(X) = INT(X*1000+.5)/1000      -Function to set output
                                           precision
 90 !
100 ! Delay
110 !
120 DEF FND                               -Exit/delay function
130 BEEP 440
140 K$=KEY$ @ IF K$="" THEN 140
150 IF UPRC$(K$)="E" THEN STOP
160 FND=0 @ END DEF
170 !
180 ! Type of dist
190 !
200 DISP CHR$(206);"ormal, ";CHR$(195);   -Determine the type of
    "hi^2, ";CHR$(198);", or ";CHR$(212    distribution to be used
    );" ";
210 INPUT A$ @ A$=UPRC$(A$[1,1])
220 IF POS("NCFT",A$)=0 THEN 200
230 DISP "Type ";CHR$(197);" at beep to
     escape" @ WAIT .7
240 IF A$="N" THEN INPUT "Enter z ?"; Z   -Input routine for normal
     @ Z=ABS(Z) @ R1=0 @ GOTO 350          distribution
250 IF A$="T" THEN INPUT "Enter t, DF ?   -Input routine for
    "; T,J @ F=T^2 @ I=1 @ GOTO 310        t-distribution
260 IF A$="C" THEN INPUT "Enter Chi^2,    -Input routine for Chi-square
    Df ?"; C,I @ F=C/I @ J=9.E400 @ GOT    distribution
    O 310
270 INPUT "Enter F, DFnum, DFden ?";F,I   -Input routine for
    ,J                                     F-distribution
280 !
290 ! Compute
300 !
310 R1=0 @ IF F<1 THEN U=I @ I=J @ J=U    -Inversion for small values
    @ F=1/F @ R1=1
320 Z=(1-2/9/J)*F^(1/3)-(1-2/9/I)         -Compute normalization
330 Z=Z/(2/9/J*F^(2/3)+2/9/I)^.5
340 IF J<3 THEN Z=Z*(1+.08*Z^4/J^3)       -Correction factor for small
                                           degrees of freedom
350 P=.5/(1+.196854*Z+.115194*Z^2+.0003   -Approximate probability using
    44*Z^3+.019527*Z^4)^4                  Jaspen's constants
360 IF R1=1 THEN P=1-P                    -Reverse inversion
370 !
380 ! Printout
390 !
400 PRINT "Probability =";FNI(P)          -Display probability
410 U=FND                                 -Call Exit/delay routine
420 GOTO 240
430 STOP
```

PROGRAM DESCRIPTION

DEPENDT - DEPENDENT (PAIRED) T-TEST

Given a set of paired observations, the t-statistic may be used to test the null hypothesis that the two samples are drawn from the same population. This program may be used to calculate the t-statistic, standard error of the difference, correlation X:Y, the mean and standard deviation of each variable, and the mean and standard deviation of the difference between the two variables.

Formulae used:

Paired T test formulas:

Number of observations = n

$$\text{Difference (D)} = Y - X$$

$$\text{Mean X} \quad (\bar{X}) = \frac{\sum_{i=1}^{n} X_i}{n} \qquad \text{Mean diff } (\bar{D}) = \frac{\sum_{i=1}^{n} (Y_i - X_i)}{n}$$

$$\text{Standard deviation of population est. } (S_x) = \left[\frac{\sum_{i=1}^{n} X_i^2}{n-1} - \frac{\left(\sum_{i=1}^{n} X_i \right)^2}{n(n-1)} \right]^{1/2}$$

$$\text{Correlation Coefficient} = \frac{S_x^2 + S_y^2 - S_{y-x}^2}{2 S_x S_y}$$

$$\text{t value} = \frac{\bar{D}}{\sqrt{S_{y-x}^2 / n}}$$

$$\text{Error of difference} = \sqrt{S_{y-x}^2 / n}$$

$$\text{Degrees of freedom} = n - 1$$

SAMPLE PROBLEM

Given a set of paired observations (below), test the null hypothesis that $V_1=V_2$. Also, compute the basic statistics of both variables.

X	14.0	17.5	17.0	17.5	15.4
Y	17.0	20.7	21.6	20.9	17.2

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "DEPNDT"	Paired T	
1		RTN to end data; D,D to delete	
	Enter data pairs	Point #1 Enter X , Y ?	14,17 [RTN]
		Point #2 Enter X , Y ?	17.5,20.7 [RTN]
		Point #3 Enter X , Y ?	17,21.6 [RTN]
	Error	Point #4 Enter X , Y ?	175,2099 [RTN]
3	Delete error	Point #5 Enter X , Y ?	D,D [RTN]
		Point #4 deleted: X=175 Y=2099	
1	Enter correct value	Point #4 Enter X , Y ?	17.5,20.9 [RTN]
	Continue	Point #5 Enter X , Y ?	15.4,17.2 [RTN]
2	Exit input routine	Point #6 Enter X , Y ?	[RTN]
4		NOTE: SD (n-1)	
	Mean and Sd(N) of X	X mean = 16.28 s = 1.53851	[RTN]
	Mean and Sd(N) of Y	Y mean = 19.48 s = 2.19932	[RTN]
	Mean and Sd(N) of difference	Diff mean = 3.2 s = 1	[RTN]
	Error of difference	Error of Diff = .44721	[RTN]
	Correlation	Correlation XY = .91676	[RTN]
	t-value and DF	t = 7.15542 DF = 4	[RTN]
5	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "DEPNDT"	Paired T	
		RTN to end data D,D to delete	
1	Enter data pair	Point #n Enter X , Y ?	X_n,Y_n [RTN]
2	To delete last pair enter D,D		
	and goto step 2	Point #n Enter X , Y ?	D,D [RTN]
3	To end data entry press [RTN]	Point #n Enter X , Y ?	[RTN]
4	Read results: Note that the		
	standard deviation is the		
	population estimate.	NOTE: SD (n-1)	
	Press [RTN] for next answer,	X mean = s =	[RTN]
	[BACK] to view previous answer	Y mean = s =	[RTN]/[BACK]
		Diff mean = s =	[RTN]/[BACK]
		Error of Diff =	[RTN]/[BACK]
		Correlation XY =	[RTN]/[BACK]
		t = DF =	[RTN]/[BACK]
5	Program options	Run again, View again, or End?	R, V, or E [RTN]
	R = run program - step 1		
	V = re-view results - step 4		
	E = end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
X( )	Observations of X	Y( )	Observations of Y
X$	X input string	Y$	Y input string
D1	Difference $\Sigma(Y-X)$	D2	$\Sigma(Y-X)^2$
X1	Sum of X ΣX	X2	Sum X squared ΣX^2
Y1	Sum of Y ΣY	Y2	Sum Y squared ΣY^2
N	Number of observations	D	Degrees of freedom
M1	Mean of X	S1	Sd of X
M2	Mean of Y	S2	Sd of Y
M3	Mean of Y-X	S3	Sd of Y-X
K$	General input string	R	Correlation X:Y
E	Error of difference	T	t value

NOTES AND REFERENCES

A maximum of 100 data pairs may be used. To change this, change the dimension statement in line 60.

The standard deviation that is computed is the population estimate (N-1).

References: Winer, B., STATISTICAL PRINCIPLES IN EXPERIMENTAL DESIGN (2nd ed., McGraw-Hill, New York 1962) page 44-49.

PROGRAM LISTING

```
 10 ! DEPNDT - Dependent
 20 ! (paired) t value
 30 ! REV 11/01/82
 40 !
 50 DISP "              Paired T" @ WAIT
    1
 60 DIM X(100),Y(100),X$[100],Y$[100]
 70 DEF FNS(S1,S2,N) = SQR(S2/(N-1)-S1^   -Function to compute standard
    2/(N*(N-1)))                           deviation (Pop. estimate)
 80 DEF FNI(X) = INT(X*100000+.5)/10000   -Function to set output
    0                                      precision
 90 !
100 ! Delay Routine
110 !
120 DEF FND                               -Wait for a key, return 1 if
                                           BACK key
130 BEEP 440
140 K$=KEY$ @ IF K$()CHR$(8) AND K$()CH
    R$(13) AND UPRC$(K$)()"E" THEN 140
150 IF UPRC$(K$)="E" THEN 620
160 FND=K$=CHR$(8)
170 END DEF
180 !
190 ! Error correction
200 !                                     -Error correction routine-
210 DEF FNE
220 FNE=1
230 IF N<1 THEN BEEP 1760 @ DISP "Must
    have data to delete " @ GOTO 290
240 DISP "Pt #";N;"deleted: X=";X(N);"Y   -Display deleted value
    =";Y(N) @ WAIT 1
250 X1=X1-X(N) @ Y1=Y1-Y(N)               -Decrement counters
260 D1=D1-Y(N)+X(N) @ D2=D2-(Y(N)-X(N))
    ^2
270 X2=X2-X(N)^2 @ Y2=Y2-Y(N)^2
280 N=N-1
290 END DEF
300 !
310 ! Enter Raw data
320 !                                     -Input routine-
330 D1,D2,X1,X2,Y1,Y2,C,N=0
340 DISP "RTN to end data  ";CHR$(196);
    " to delete"; @ WAIT 1 @ ON ERROR G
    OTO 380
350 DISP "Point #";N+1;" Enter X , Y ";
360 INPUT X$,Y$ @ IF X$="" THEN 460       -Enter value and check for
                                           escape
370 IF POS("DdEe",X$[1,1) THEN U=FNE @    -Call error correction routine
     GOTO 350 ELSE GOTO 390
380 IF ERRN=81 THEN 460 ELSE GOTO 350
390 N=N+1 @ X(N)=VAL(X$) @ Y(N)=VAL(Y$)   -Increment counters
400 D1=D1+Y(N)-X(N) @ D2=D2+(Y(N)-X(N))
    ^2
410 X1=X1+X(N) @ X2=X2+X(N)^2 @ Y1=Y1+Y
    (N) @ Y2=Y2+Y(N)^2
```

PROGRAM LISTING

```
420 GOTO 350
430 !
440 ! Compute mean,std dev
450 !
460 IF N<2 THEN BEEP 1760 @ DISP "Must
    have more than one point" @ WAIT 1
    @ GOTO 350
470 OFF ERROR @ M1=X1/N @ M2=Y1/N @ S1=
    FNS(X1,X2,N) @ S2=FNS(Y1,Y2,N) @ D=
    N-1
480 M3=D1/N @ S3=FNS(D1,D2,N)

490 R=(S1^2+S2^2-S3^2)/(2*S1*S2)

500 T=M3/SQR(S3^2/N) @ E=SQR(S3^2/N)

510 PRINT "NOTE : SD.=n-1"
520 PRINT "X mean=";FNI(M1);" s=";FNI(S
    1) @ U=FND @ IF U THEN 510
530 PRINT "Y mean=";FNI(M2);" s=";FNI(S
    2)
540 U=FND @ IF U THEN 520
550 PRINT "Diff mean=";FNI(M3);" s=";FN
    I(S3) @ U=FND @ IF U THEN 530
560 PRINT "Error of diff. =";FNI(E) @ U
    =FND @ IF U THEN 550
570 PRINT "Correlation XY =";FNI(R) @ U
    =FND @ IF U THEN 560
580 PRINT "t=";FNI(T);"  DF=";D @ U=FND
     @ IF U THEN 570
590 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);"nd";
600 INPUT X$ @ X$=UPRC$(X$&" ")
610 ON POS("RVE",X$[1,1])+1 GOTO 590,33
    0,510,620
620 END
```

-Calculate the mean and SD of X and Y

-Calculate the mean and Sd of Y-X

-Calculate the correlation coefficient

-Calculate the t value and the error of the difference

-Printout

-Program options

PROGRAM DESCRIPTION

INDPT - t TEST FOR 2 UNEQUAL SIZED SAMPLES

If two independent samples of size N1 and N2 are drawn at random from a normally distributed population with a mean $\bar{x}$ and a variance of σ^2, this statistic is used to test whether or not the two samples have been drawn from the same normal population or from normal populations with different means but the same variance. This program will calculate the t statistic and the basic statistics for a set of data entered by keyboard.

Formulae:

$$\text{Mean 1 } (\bar{X}) = \frac{\sum_{i=1}^{n_x} X_i}{n_x} \qquad \text{Mean 2 } (\bar{Y}) = \frac{\sum_{i=1}^{n_y} Y_i}{n_y}$$

$$\text{Numerator of standard deviation } (L_x) = n_x \sum_{i=1}^{n_x} X_i^2 - \left(\sum_{i=1}^{n_x} X_i \right)^2$$

$$\text{Standard deviation (population est.)} = \sqrt{\frac{L_x}{n_x(n_x - 1)}}$$

$$t = \sqrt{\frac{(n_x + n_y - 2)\left(n_y \sum_{i=1}^{n_x} X_i - n_x \sum_{i=1}^{n_y} Y_i \right)^2}{(n_x + n_y)(n_y L_x + n_x L_y)}}$$

$$\text{If } \bar{X} - \bar{Y} \neq 0 \quad \text{then} \quad t = t * \text{SGN}(\bar{X} - \bar{Y})$$

$$\text{where: } \text{SGN}(V) = \frac{V}{\sqrt{V^2}} \; ; \quad V \neq 0$$

$$\text{Degrees of freedom (df)} = (n_x - 1)(n_y - 1)$$

SAMPLE PROBLEM

Calculate the basic statistics for the 2 unequal sized variables found below, and determine whether or not they have been drawn from the same population (calculate the t value).

Ident.	1	2	3	4	5	6	7	8	9	10
X	79	84	108	114	120	103	122	120		
Y	91	103	90	113	108	87	100	80	99	54

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "INDPT"	Independent Sample t test	
		RTN = finish treat, D = delete	
1	Enter X treatment	Trt 1 Point 1 ?	79 [RTN]
		Trt 1 Point 2 ?	84 [RTN]
		Trt 1 Point 3 ?	108 [RTN]
		Trt 1 Point 4 ?	114 [RTN]
		Trt 1 Point 5 ?	120 [RTN]
		Trt 1 Point 6 ?	103 [RTN]
		Trt 1 Point 7 ?	122 [RTN]
		Trt 1 Point 8 ?	120 [RTN]
	End X treatment	Trt 1 Point 9 ?	[RTN]
2	Enter Y treatment	Trt 2 Point 1 ?	91 [RTN]
		Trt 2 Point 2 ?	103 [RTN]
		Trt 2 Point 3 ?	90 [RTN]
		Trt 2 Point 4 ?	113 [RTN]
		Trt 2 Point 5 ?	108 [RTN]
	Error	Trt 2 Point 6 ?	888 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
4	Delete last input	Trt 2 Point 7 ?	D [RTN]
	Last value is deleted	Point 6 Deleted = 888	
2	Enter correct value	Trt 2 Point 6 ?	87 [RTN]
		Trt 2 Point 7 ?	100 [RTN]
		Trt 2 Point 8 ?	80 [RTN]
		Trt 2 Point 9 ?	99 [RTN]
		Trt 2 Point 10 ?	54 [RTN]
	End Y treatment	Trt 2 Point 11 ?	[RTN]
5	NOTE: Sd=Population est.	NOTE: Sd/n-1	
	Mean and Sd. of X	Mean 1 = 106.25 s 1 = 16.637	[RTN]
	Mean and Sd. of Y	Mean 2 = 92.50 s 2 = 16.821	[RTN]
	t value	t = 1.732	[RTN]
	Degrees of freedom	DF = 16	[RTN]
6	Review routine	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' then re-run program		
	If 'V' then re-view results		
	If 'E' then stop		

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "INDPT"	Independent Sample t test	
		RTN to finish treat, D to delete	
1	Enter X as prompted or	Trt 1 Point n ?	Xn [RTN]
3	End treatment and goto step 2	Trt 1 Point n ?	[RTN] or
4	Delete last input	Trt 1 Point n ?	D [RTN]
	Value will be deleted	Point n deleted = Xn	
	Goto step 1 until done		
2	Enter Y treatment. Apply		
	steps 3 or 4 as above	Trt 2 Point n ?	Yn [RTN]
5	Read output	NOTE: SD/n-1	
	[RTN] = view next output	Mean 1 = s 1 =	[RTN]
	[BACK] = review last output	Mean 2 = s 2 =	[RTN]/[BACK]
	REMEMBER: In this program	t =	[RTN]/[BACK]
	the standard deviation is	DF =	[RTN]/[BACK]
	for the population estimate		
6	Review routine	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' then re-run program		
	If 'V' then re-view output		
	If 'E' then stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
N	Input counter	D	Total number of observations
XO	N of obs. in X n_x	YO	N of obs. in Y n_y
X1	Sum x = Σx	Y1	Sum y = Σy
X2	Sum x squared = Σx^2	Y2	Sum y squared = Σy^2
K$	User input string	A( )	Temporary data storage
L1	$(n_x)(\Sigma x^2)-(\Sigma x)^2$	Y	Index counter
L2	$(n_y)(\Sigma y^2)-(\Sigma y)^2$	S1	Standard dev. of X (n-1)
S2	Standard dev. of Y	T	t value
M1	Mean of x	M2	Mean of y

NOTES AND REFERENCES

This program will accept a maximum of 100 data per treatment. Change the dimension statements in line 90 to change this maximum.

The standard deviation in this program is calculated for the population estimate.

References: Winer, B.J., STATISTICAL PRINCIPLES IN EXPERIMENTAL DESIGN 2nd edition, (McGraw-Hill, New York, 1971) pages 35-37.

PROGRAM LISTING

```
 10 ! INDPT - Independent
 20 ! Sample t-test
 30 ! REV 11/01/82
 40 !
 50 DISP "  Independent Sample t test"
    @ WAIT 1
 60 !
 70 ! Initialize
 80 !
 90 DIM K$[100],A(100)
100 K$=""
110 DEF FNI(X) = INT(X*1000+.5)/1000      -Define the output precision
120 !
130 ! Delay routine
140 !
150 DEF FND                               -Wait for a key- return 1 for
                                           the BACK key
160 BEEP 440
170 K$=KEY$ @ IF K$<>CHR$(13) AND K$<>C
    HR$(8) THEN 170
180 FND=K$=CHR$(8)
190 END DEF
200 !
210 ! Correction
220 !
230 DEF FNE                               -Error correction
240 IF N<1 THEN BEEP @ DISP "Must have
    data to delete" @ WAIT 1 @ GOTO 280
250 IF Y=1 THEN X0=X0-1 @ X1=X1-A(N) @
    X2=X2-A(N)^2 ELSE Y0=Y0-1 @ Y1=Y1-A
    (N) @ Y2=Y2-A(N)^2
260 DISP "Point";N;"Deleted = ";A(N)
270 N=N-1 @ D=D-1
280 FNE=0 @ END DEF
290 !
300 ! Enter data
310 !
320 DISP "RTN = Finish treat.  D=delete
    " @ WAIT 1
330 N,D,X0,X1,X2,Y0,Y1,Y2=0
340 ON ERROR BEEP @ DISP "Invalid data    -Error trapping for invalid
    - please retype" @ GOTO 360            input
350 FOR Y=1 TO 2                          -Input loop-
360 DISP "Trt.";Y;"Point";N+1; @ INPUT
    K$
370 IF K$="" THEN 450                     -Call end of treatment
380 IF POS("DdEe",UPRC$(K$[1,1])) THEN    -Call error correction
    U=FNE @ GOTO 360
390 N=N+1 @ D=D+1 @ A(N)=VAL(K$)
400 IF Y=1 THEN X0=X0+1 @ X1=X1+A(N) @    -Increment counters
    X2=X2+A(N)^2 ELSE Y0=Y0+1 @ Y1=Y1+A
    (N) @ Y2=Y2+A(N)^2
410 GOTO 360
420 !
```

PROGRAM LISTING

```
430 ! End treatment
440 !
450 IF N<2 THEN DISP "Must have more th
    an 1 point" @ BEEP @ WAIT 1 @ GOTO
    360
460 N=0
470 NEXT Y
480 !
490 ! Compute
500 !
510 L1=X0*X2-X1^2 @ L2=Y0*Y2-Y1^2          -Compute numerator of standard
                                            deviation
520 S1=SQR(L1/(X0*(X0-1))) @ S2=SQR(L2/     -Compute rest of standard
    (Y0*(Y0-1)))                            deviation
530 M1=X1/X0 @ M2=Y1/Y0                     -Compute the mean of X and Y
540 T=SQR((X0+Y0-2)*(Y0*X1-X0*Y1)^2/((X     -Compute t value
    0+Y0)*(Y0*L1+X0*L2)))
550 IF M1-M2<>0 THEN T=T*SGN(M1-M2)         -Multiply by the sign of mean X
                                            minus mean Y
560 D=X0-1+(Y0-1)                           -Compute the degrees of freedom
570 !
580 ! Print-out
590 !
600 PRINT "NOTE: SD/n-1"                    -Printout and call delay
                                            routine
610 PRINT "Mean 1 =";FNI(M1);"  s 1=";F
    NI(S1) @ U=FND @ IF U THEN 600
620 PRINT "Mean 2 =";FNI(M2);"  s 2 =";
    FNI(S2) @ U=FND @ IF U THEN 610
630 PRINT "t =";FNI(T) @ U=FND @ IF U T
    HEN 620
640 PRINT "DF =";D @ U=FND @ IF U THEN
    630
650 !
660 ! End dialogue
670 !
680 DISP CHR$(210);"un again, ";CHR$(21     -Continuation options
    4);"iew again, or ";CHR$(197);"nd";
690 INPUT X$ @ X$=UPRC$(X$&" ")
700 ON POS("RVE",X$[1,1])+1 GOTO 680,32
    0,600,710
710 STOP
```

PROGRAM DESCRIPTION

CHISQR - CHI-SQUARE TEST

Contingency tables are used to test the null hypothesis that two variables are independent. Pearson's coefficient of contingency measures the degree of association between the two variables. The Chi-square statistic is computed with the assumption that the samples are independent and have been entered in nominal form (not as percentages).

Formulae:

$$\chi^2 = \sum_{i=1}^{r}\sum_{j=1}^{k}\frac{(O_{ij} - E_{ij})^2}{E_{ij}}$$

Where

O_{ij} = Observed number of cases categorized in the i^{th} row of the j^{th} column.

$$\text{Expected value } (E_{ij}) = \frac{\sum_{a=1}^{r} A_{aj} \quad \sum_{b=1}^{k} A_{ib}}{\sum_{c=1}^{r}\sum_{d=1}^{k} A_{cd}}$$

r = the number of rows in the data set
k = the number of columns in the data set

$$\text{Contingency coefficient } (C) = \sqrt{\frac{\chi^2}{N + \chi^2}}$$

Where

$$N = \sum_{i=1}^{r}\sum_{j=1}^{k} A_{ij}$$

Degrees of freedom (df) = $(r - 1)(k - 1)$

SAMPLE PROBLEM

Find the Test Statistic Chi-Square for the following set of data:

	1	2	3	4
1	36	67	49	58
2	31	60	49	54
3	58	87	80	68

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "CHISQR"	Chi-Square test	
1	Enter matrix size: R	How many rows:	3 [RTN]
2		How many columns?	4 [RTN]
3.1	Enter A(1,1)	Row 1 Column 1 ?	36 [RTN]
	Enter A(1,2)	Row 1 Column 2 ?	67 [RTN]
	Enter A(1,3)	Row 1 Column 3 ?	49 [RTN]
	Enter A(1,4)	Row 1 Column 4 ?	58 [RTN]
3.2	Enter A(2,1)	Row 2 Column 1 ?	31 [RTN]
	Enter A(2,2)	Row 2 Column 2 ?	60 [RTN]
	Enter A(2,3)	Row 2 Column 3 ?	49 [RTN]
	Enter A(2,4)	Row 2 Column 4 ?	54 [RTN]
3.3	Enter A(3,1)	Row 3 Column 1 ?	58 [RTN]
	Error	Row 3 Column 2 ?	887 [RTN]
	Enter A(3,3)	Row 3 Column 3 ?	80 [RTN]
	Enter A(3,4)	Row 3 Column 4 ?	68 [RTN]
4	Call error correction	OK? ([RTN]=continue N=change)	N [RTN]
	Enter index of incorrect value	Enter Row #, Column # ?	3,2 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Enter correct value	Old value: 887 New?	87 [RTN]
	Exit	OK? ([RTN]=continue N=change)	[RTN]
5	Read observed and expected	1,1 O=36 E=37.6614	[RTN]
	values	1,2 O=67 E=64.4763	[RTN]
		1,3 O=49 E=53.6298	[RTN]
		1,4 O=58 E=54.2324	[RTN]
		2,1 O=31 E=34.792	[RTN]
		2,2 O=60 E=59.5638	[RTN]
		2,3 O=49 E=49.5438	[RTN]
		2,4 O=54 E=50.1005	[RTN]
		3,1 O=58 E=52.5466	[RTN]
		3,2 O=87 E=89.9598	[RTN]
		3,3 O=80 E=74.8264	[RTN]
		3,4 O=68 E=75.6671	[RTN]
	View Chi-square value	Chi-Square = 3.3574	[RTN]
	View degrees of freedom value	Degrees of Freedom = 6	[RTN]
	View contingency coefficient	Contingency coeff = .0692	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "CHISQR"	Chi-Square test	
1	Enter the number of rows	How many rows?	N [RTN]
2	Enter the number of columns	How many columns?	C [RTN]
3	Enter the data as prompted	Row_i $Column_j$	A_{ij} [RTN]
4	Check for incorrect value	OK? ([RTN]=continue N=change)	N [RTN]
	If values are correct goto 5		
		Enter Row #, Column #	Re, Ce [RTN]
		Old value = A_{RC} New?	Correct value [RTN]
	Goto 4		
5	View observed and expected	(r,c) O= E=	[RTN]/[BACK]
	values for all input data.		
	Use [RTN] to continue, [BACK]		
	to review.		
	View Chi-square value	Chi-square =	[RTN]/[BACK]
	View degrees of freedom value	Degrees of Freedom =	[RTN]/[BACK]
	View contingency coefficient	Contingency Coeff =	[RTN]/[BACK]
6	Continuation options	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' then re-run		
	If 'V' then re-view		
	If 'E' then stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A(,)	Data storage matrix	A$	Input string
R(j)	Sum of row j	R	Number of rows
C(j)	Sum of column j	C	Number of columns
E(,)	Expected value	T	Sum of all data
C2(,)	Chi-square value of datum	I,J	Index
CO	Contingency coefficient	C2	Total chi-square
U	Delay: 1=[BACK] key used	D	Degrees of freedom
R1, C1	Error correction index		

NOTES AND REFERENCES

A maximum of 15 subjects and 15 variables may be entered. To change this, change the dimension statements in line 60.

References: Siegel, Sidney, NON PARAMETRIC STATISTICS FOR THE BEHAVORIAL SCIENCES (McGraw Hill, New York, 1956) page 197, 104-107.

Conover, W.J., PRACTICAL NON PARAMETRIC STATISTICS (John Wiley, New York, 1971)

PROGRAM LISTING

```
 10 ! CHISQR - Chi
 20 ! Square test
 30 ! REV 11/01/82
 40 !
 50 DISP "    Chi-Square test" @ WAIT 1
 60 DIM A(15,15),R(15),C(15),E(15,15),C
    2(15,25),A$[100]
 70 T=0 @ C2=0
 80 !
 90 ! Short Prec
100 !
110 DEF FNI(K) = INT(K*10000+.5)/10000    -Set precision of output
120 !
130 ! Delay routine
140 !
150 DEF FND                               -Wait for a key- return 1 if
                                           BACK key
160 BEEP 440
170 A$=KEY$ @ IF A$<>CHR$(13) AND A$<>C
    HR$(8) THEN 170
180 FND=A$=CHR$(8)
190 END DEF
200 !
210 ! Start dialogue
220 !
230 INPUT "How many rows ?";R
240 INPUT "How many Columns ?";C
250 IF R<1 OR C<1 THEN BEEP @ RUN
260 !
270 ! Input loop
280 !                                     -Data entry routine
290 FOR I=1 TO R @ R(I)=0                 -Row loop
300 FOR J=1 TO C @ C(J)=0                 -Column loop
310 DISP "Row";I;"Column";J;              -Display index and input value
320 INPUT A(I,J) @ NEXT J @ NEXT I
330 !
340 ! Error correction
350 !
360 INPUT "Ok? (RTN=continue, N=change)   -Error correction routine
     ?"; A$ @ A$=A$&" "
370 IF UPRC$(A$[1,1])<>"N" THEN 420       -Check for error
380 INPUT "Enter Row# , Clm# ?";R1,C1     -Enter the index value of the
                                           error
390 IF R1<1 OR R1>R OR C1<1 OR C1>C THE   -Check validity of data
    N BEEP 880 @ DISP "Impossible datum
    " @ GOTO 360
400 DISP "Old value:";A(R1,C1);"  New";   -Enter new value
     @ INPUT A(R1,C1)
410 GOTO 360
420 FOR I=1 TO R @ FOR J=1 TO C           -Summation loop-
430 R(I)=R(I)+A(I,J)                      -Sum rows
440 C(J)=C(J)+A(I,J)                      -Sum columns
450 T=T+A(I,J)                            -Sum all values
460 NEXT J @ NEXT I
```

PROGRAM LISTING

```
470 FOR I=1 TO R @ FOR J=1 TO C                  -Computation loop-
480 E(I,J)=R(I)*C(J)/T                           -Compute expected value
490 C2(I,J)=(A(I,J)-E(I,J))^2/E(I,J)             -Compute chi-square
                                                  contribution
500 C2=C2+C2(I,J)
510 NEXT J @ NEXT I
520 D=(C-1)*(R-1)                                -Compute the degrees of freedom
530 C0=SQR(C2/(T+C2))                            -Compute contingency
                                                  coefficient
540 !
550 ! Print-out
560 !
570 FOR I=1 TO R @ FOR J=1 TO C                  -Print the observed and
                                                  expected values
580 A$=STR$(I)&","&STR$(J)
590 PRINT A$;"  O=";A(I,J);"  E=";FNI(E
    (I,J))
600 U=FND @ IF U AND J>1 THEN J=J-2
610 IF U=1 AND J=1 AND I>1 THEN I=I-2 @
     GOTO 630
620 NEXT J
630 NEXT I
640 PRINT "Chi-Square =";FNI(C2) @ U=FN          -Print the chi-square value,
    D @ IF U THEN 570                             Df, and contingency coef.
650 PRINT "Degrees of Freedom =";D @ U=
    FND @ IF U THEN 640
660 PRINT "Contingency Coeff =";FNI(C0)
670 U=FND @ IF U THEN 650
680 DISP CHR$(210);"un again, ";CHR$(21          -Continuation options
    4);"iew again, or ";CHR$(197);
690 INPUT "nd ?"; A$ @ A$=UPRC$(A$&" ")
700 ON POS("RVE",A$[1,1])+1 GOTO 680,23
    0,570,710
710 END
```

PROGRAM DESCRIPTION

ONE-WAY ANALYSIS OF VARIANCE

The one-way analysis of variance is used to test whether observed differences among sample means can be attributed to chance or whether they indicate actual differences among the corresponding population means. The F ratio is the test statistic for determining if the null hypothesis can be rejected at a given level. The "between group variance" is compared with "within group variance". The R-squared is a measure of the relationship between the independent and dependent variables in the sample in the fixed effect case. Omega square is a measure of the independent-dependent variable association in the population in the fixed effect case. This program will compute an analysis of variance table, F-ratio, R-squared and Omega squared for a set of data input by the user.

Formulae: n = number of samples

K = number of treatments

Sum of column $j = T_j = \sum_{i=1}^{n} X_i$

Mean of observations in the i^{th} sample $= \bar{X}_i = \dfrac{\sum_{i=1}^{n_i} X_i}{n_i}$

Standard deviation of observation in the i^{th} sample:

$$S_i = \left[\frac{\sum_{i=1}^{n_i} X_{ij} - \bar{X}_i^2}{n_i} \right]^{1/2}$$

Sum of squares (SS):

$$\text{SS Total} = \sum_{i=1}^{n}\sum_{j=1}^{n} X_{ij}^2 - \frac{\left(\sum_{i=1}^{n}\sum_{j=1}^{n} X_{ij}\right)^2}{n}$$

$$\text{SS Treatment} = \sum_{j=1}^{n} \frac{T_j^2}{n_j} - \frac{\left(\sum_{i=1}^{n}\sum_{j=1}^{n} X_{ij}\right)^2}{n}$$

$$\text{SS Error} = \sum_{j=1}^{n} \sum_{j=1}^{n} X_j^2 - \sum_{j=1}^{n} \frac{T_j^2}{n_j}$$

PROGRAM DESCRIPTION

ONE-WAY ANALYSIS OF VARIANCE (continued)

Treatment degrees of freedom: $df_1 = K - 1$

Error degrees of freedom: $df_2 = N - K$

Total degrees of freedom: $df_3 = N - 1$

Mean square treatment = SS Treatment/df_1

Mean square error = SS Error/df_2

$$\text{The F ratio} = \frac{\text{Mean square treatment}}{\text{Mean square error}}$$

$$R^2 = \frac{\text{SS treatment}}{\text{SS total}}$$

$$\text{Omega}^2 = \frac{\text{SS treatment} - (df_1)(\text{Mean square error})}{\text{SS total} + \text{Mean square error}}$$

SAMPLE PROBLEM

The following are the scores obtained in an achievement test by a random sample of students from four different schools:

		Student						
		1	2	3	4	5	6	7
School	1	88	99	96	68	85		
	2	78	62	98	83	61	88	
	3	80	61	74	92	78	54	77
	4	71	65	90	46			

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "ONEAOV"	One Way Analysis of Variance	
1	Enter the # of treatments	How many treatments?	4 [RTN]
	Push [RTN] to end treatments	Depress [RTN] to end treatment	
2	Begin treatment 1	Treat 1 Subject 1 ?	88 [RTN]
		Treat 1 Subject 2 ?	99 [RTN]
		Treat 1 Subject 3 ?	96 [RTN]
		Treat 1 Subject 4 ?	68 [RTN]
		Treat 1 Subject 5 ?	85 [RTN]
4	End treatment 1	Treat 1 Subject 6 ?	[RTN]
2	Begin treatment 2	Treat 2 Subject 1 ?	78 [RTN]
		Treat 2 Subject 2 ?	62 [RTN]
		Treat 2 Subject 3 ?	98 [RTN]
		Treat 2 Subject 4 ?	83 [RTN]
		Treat 2 Subject 5 ?	61 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Treat 2 Subject 6 ?	88 [RTN]
4	End treatment 2	Treat 2 Subject 7 ?	[RTN]
2	Begin treatment 3	Treat 3 Subject 1 ?	80 [RTN]
		Treat 3 Subject 2 ?	61 [RTN]
		Treat 3 Subject 3 ?	74 [RTN]
		Treat 3 Subject 4 ?	92 [RTN]
		Treat 3 Subject 5 ?	78 [RTN]
		Treat 3 Subject 6 ?	54 [RTN]
		Treat 3 Subject 7 ?	77 [RTN]
4	End treatment 3	Treat 3 Subject 8 ?	[RTN]
2	Begin treatment 4	Treat 4 Subject 1 ?	71 [RTN]
		Treat 4 Subject 2 ?	65 [RTN]
		Treat 4 Subject 3 ?	90 [RTN]
	Error	Treat 4 Subject 4 ?	496 [RTN]
3	Delete last entry	Treat 4 Subject 5 ?	E [RTN]
	Incorrect value deleted	(4,4) Deleted = 496	
2	Enter correct value	Treat 4 Subject 4?	46 [RTN]
2	End treatment 4	Treat 4 Subject 5 ?	[RTN]
5	Read mean and standard	Trt 1 Mean = 87.2 s = 10.87	[RTN]
	deviation	Trt 2 Mean = 78.333 s = 13.35	[RTN]
		Trt 3 Mean = 73.714 s = 11.671	[RTN]
		Trt 4 Mean = 68 s = 15.7	[RTN]
	Total mean and standard		
	deviation	Total Mean 77 s = 14.35	[RTN]
	Treatment SS, Df	Treat SS = 930.438 Df = 3	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Treatment MS	Treat MS = 310.146	[RTN]
	Error SS, Df	Error SS = 3599.562 Df = 18	[RTN]
	Error MS	Error MS = 199.976	
	Total SS, Df	Total SS = 4530 Df = 21	[RTN]
	F value	F = 1.55092	[RTN]
	R squared	R squared = .20539	[RTN]
	Omega squared	Omega squared = .06988	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "ONEAOV"	One Way Analysis of Variance	
1	Enter the number of treatments	How many treatments?	k [RTN]
2		Depress RTN to end treatment	
2.1	Enter data item	Treat. Subject ?	A_{ij} [RTN]
3	To correct previous data item:	Treat. Subject ?	E [RTN]
		(i,j-1) DELETED = A_{ij-1}	
4	If all subjects for the treat-		
	ment have been entered, goto		
	step 5	Treat. Subject ?	[RTN]
5	Read mean and standard devi-		
	ation for each treatment	Treat n Mean = s =	[RTN]
6	View results: press [RTN] to	Total Mean = s =	[RTN]
	see next answer, [BACK] to see	Treat SS = Df =	[RTN]/[BACK]
	previous answer	Treat MS =	[RTN]/[BACK]
		Error SS = Df =	[RTN]/[BACK]
		Error MS =	[RTN]/[BACK]
		Total SS = Df =	[RTN]/[BACK]
		F =	[RTN]/[BACK]
		R squared =	[RTN]/[BACK]
		Omega squared =	[RTN]/[BACK]
7	Continuation options:	Run again, View again, or End?	R, V, or E [RTN]
	If 'R' then step 1		
	If 'V' then step 6		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
X$	General string input	X	Value of data point
J	Column counter	K	Number of columns
N	Total number of subjects	N(1)	Number of subjects per column
T	Sum of all data Σx	T( )	Sum of columns Σx_j
X2	Sum square of data set $\Sigma(x2)$	X2( )	Sum square of columns $\Sigma(x^2)$
S9	Standard deviation of data	S9( )	Standard deviation of column
M	Mean of all data	M( )	Mean of columns
V	Variance of data	V( )	Variance of columns
IØ	Intermediate step $\Sigma \frac{T_j^2}{n_j}$	IØ( )	Intermediate step $\frac{T_J^2}{n_j}$
Q( )	Sum square of columns SS_τ	S1	SS treatment
S2	SS error	S3	SS total
M1	MS treatment	M2	MS error
II	Intermediate step $\frac{(\Sigma x)^2}{n}$	D1	Degrees of freedon : treatment
D2	Degrees of freedom : error	D3	Degrees of freedom : total
F	F ratio	R2	R squared R^2
O2	Omega square Ω^2	E1	Value to delete from counters
FNI(x)	Short precision output	FNJ(x)	Long precision output
FND( )	Delay: 1=last line	FNS( )	Compute variance
R( )	Temporary storage of trt.		

NOTES AND REFERENCES

A maximum of 100 observations per treatment may be used. To change this, change lines 80 and 560.

A maximum of 20 treatments may be used. To change this, change lines 80 and 440.

Once a treatment has been finished by the solo [RTN], it may not be edited.

References: Winer, B.J., STATISTICAL PRINCIPLES IN EXPERIMENTAL DESIGN (McGraw-Hill, New York; 1971) page 210-214.

Hays, W.L., STATISTICS FOR PSYCHOLOGISTS (Holt, Rinehart and Winston; 1963) page 382.

PROGRAM LISTING

```
 10 ! ONEAOV - One way
 20 ! Analysis of Variance
 30 ! Rev 11/01/82
 40 !
 50 DELAY .5
 60 DISP 'One Way Analysis of Variance'
 70 REAL N,T,X2,V,M,S1,S2,S3,M1,M2,M3,D
    1,D2,D3,S9,F,R2,O2                    -Define variables and set up
 80 REAL N(20),T(20),X2(20),I0(20),Q(20     array tables
    ),V(20),M(20),S9(20),R(100)
 90 DIM X$[20]
100 !
110 ! Define variance
120 !
130 DEF FNS(S1,S2,N) = S2/N-(S1/N)^2      -Function to compute the
                                           variance
140 !
150 ! Delay routine
160 !
170 DEF FND                               -Delay until key pressed,
                                           return 1 for BACK key
180 BEEP 440,.1
190 A$=KEY$ @ IF A$<>CHR$(13) AND A$<>C
    HR$(8) THEN 190
200 FND=0 @ IF A$=CHR$(8) THEN FND=1
210 END DEF
220 !
230 ! Integer routine
240 !
250 DEF FNI(X) = INT(X*1000+.5)/1000      -Integer function for short
                                           output precision
260 DEF FNJ(X) = INT(X*100000+.5)/10000   -Integer function for long
    0                                      output precision
270 !
280 ! Correction
290 !
300 DEF FNE(C)                            -Error correction - delete last
                                           input from the counters
310 IF N(C)<1 THEN BEEP @ DISP "Must ha
    ve data to delete" @ GOTO 350
320 E1=R(N(C)) @ DISP "(";C;",";N(C);")
     DELETED =";R(N(C)) @ WAIT 1
330 T(C)=T(C)-E1 @ T=T-E1 @ N(C)=N(C)-1
     @ N=N-1
340 X2=X2-E1^2 @ X2(C)=X2(C)-E1^2
350 FNE=0 @ END DEF
360 !
370 ! Initialize
380 !
390 FOR Y=1 TO 20 @ N(Y)=0 @ T(Y)=0 @ X   -Set starting values of all
    2(Y)=0 @ I0(Y)=0 @ Q(Y)=0 @ V(Y)=0     counters to zero
    @ M(Y)=0 @ NEXT Y
400 N,T,X2,I0,Q=0
410 !
```

PROGRAM LISTING

```
420 ! Enter data
430 !
440 INPUT "How many Treatments ?"; K @
    IF K<2 OR K>20 THEN BEEP @ GOTO 440
450 DISP "Depress RTN to end treatment"
     @ BEEP 220 @ WAIT .9
460 FOR J=1 TO K
470 DISP 'Treat.';J;'Subject';N(J)+1; @
     INPUT X$
480 IF X$="" THEN 610
490 IF POS("DdEe",X$) THEN U=FNE(J) @ G
    OTO 470
500 X=VAL(X$)
510 !
520 ! Loop counters
530 !
540 N(J)=N(J)+1 @ N=N+1 @ R(N(J))=X
550 T(J)=T(J)+X @ T=T+X @ X2(J)=X2(J)+X
    ^2 @ X2=X2+X^2
560 IF N(J)>=100 THEN BEEP @ DISP "Can'
    t Enter any more" @ WAIT 1 @ GOTO 6
    10
570 GOTO 470
580 !
590 ! End of treatment
600 !
610 BEEP 220 @ IF N(J)<2 THEN BEEP @ DI
    SP "Must have more than one subject
    " @ GOTO 470
620 NEXT J
630 !
640 ! CALCULATION
650 ! Me,STD,Internal
660 !
670 FOR J=1 TO K
680 I0(J)=T(J)^2/N(J) @ I0=I0+I0(J)
690 Q(J)=X2(J)-I0(J) @ Q=Q+Q(J)
700 V(J)=FNS(T(J),X2(J),N(J)) @ S9(J)=S
    QR(V(J))
710 M(J)=T(J)/N(J)
720 PRINT "Trt";J;"Mean=";FNI(M(J));" s
    =";FNI(S9(J)) @ U=FND
730 NEXT J @ V=FNS(T,X2,N) @ M=T/N @ S9
    =SQR(V)
740 PRINT "Total  Mean=";FNI(M);" s=";F
    NI(S9) @ U=FND
750 !
760 ! Int,SS,MS,DF
770 !
780 I1=T^2/N
```

-Enter the number of treatments

-Beginning of loop to enter treatments

-Enter a string to be analyzed

-If null exit loop and goto next item. If D or E call FNE

-Convert string value to numeric

-Increment counters

-Ensure that more than one subject has been entered

-Loop to compute mean and Sd of each treatment

-Compute intermediate sum

-Compute SS of columns

-Compute variance and standard deviation

PROGRAM LISTING

```
 790 S1=I0-I1 @ S2=X2-I0 @ S3=X2-I1          -Compute the sum of squares
 800 D1=K-1 @ D2=N-K @ D3=N-1                -Compute the degrees of freedom
 810 M1=S1/D1 @ M2=S2/D2                     -Compute the mean squares
 820 !
 830 ! F,R^2,O^2
 840 !
 850 F=M1/M2 @ R2=S1/S3 @ O2=(S1-(K-1)*M     -Compute F, R-squared,
     2)/(S3+M2)                               omega-squared
 860 !
 870 ! Print Out
 880 !
 890 PRINT "Treat SS=";FNI(S1);"Df=";D1      -Print out results
     @ U=FND @ IF U THEN 740
 900 PRINT "Treat  MS=";FNI(M1) @ U=FND
     @ IF U THEN 890
 910 PRINT "Error SS=";FNI(S2);"Df=";D2
     @ U=FND @ IF U THEN 900
 920 PRINT "Error  MS=";FNI(M2) @ U=FND
     @ IF U THEN 910
 930 PRINT "Total SS=";FNI(S3);"Df=";D3
     @ U=FND @ IF U THEN 920
 940 PRINT "F=";FNJ(F) @ U=FND @ IF U TH
     EN 930
 950 PRINT "R squared=";FNJ(R2) @ U=FND
     @ IF U THEN 940
 960 PRINT "Omega squared=";FNJ(O2) @ U=
     FND @ IF U THEN 950
 970 DISP CHR$(210);"un again, ";CHR$(21     -Continuation options
     4);"iew again, or ";CHR$(197);"nd";
 980 INPUT X$ @ ON POS("RVE",UPRC$(X$))+
     1 GOTO 970,390,740,990
 990 DISP
1000 STOP
```

PROGRAM DESCRIPTION

PREDICT - SIMPLE LINEAR REGRESSION

This program calculates (or accepts) the basic statistics that are used to describe two variables X and Y, and uses that information to calculate the F value, test the correlation against zero, produce the slope and intercept of the regression line, determine the standard estimate of Y when predicted from X, and determine the 95% confidence range of Y.

Formula:

Formula (Linear Regression Prediction)

$$B_{xy} = \hat{B}_1 = r_{xy}\frac{S_y}{S_x}$$

$$A_{xy} = \hat{B}_0 = \bar{Y} - (\hat{B}_1)\bar{X}$$

$$SE = S_y(\sqrt{1 - r_{xy}^2})$$

$$\hat{Y} = XB_{yx} + A_{yx}$$

Correlation coefficient:

$$r = \frac{\sum_{i=1}^{n} X_i Y_i - \left(\sum_{i=1}^{n} X_i\right)\left(\sum_{i=1}^{n} Y_i\right)\Big/ n}{\left\{\left[\sum_{i=1}^{n} X_i^2 - \left(\sum_{i=1}^{n} X_i\right)^2\Big/ n\right]\left[\sum_{i=1}^{n} Y_i^2 - \left(\sum_{i=1}^{n} Y_i\right)^2\Big/ n\right]\right\}^{1/2}}$$

f value: $f = (r_{xy})^2/((1 - (r_{xy})^2)/(n - 2))$

Slope: $B_{xy} = r_{xy}(Sd_y/Sd_x)$

y intercept: $A_{xy} = \bar{Y} - (B_{xy})(\bar{X})$

Standard error: $Se = \pm(Sd_y)(\sqrt{1 - r_{xy}^2})$

y predicted from x: $\hat{Y} = XB_{yx} + A_{yx}$

SAMPLE PROBLEM

Below are the scores achieved by students on a math test and a reading test:

Achieved math:	70	75	80	80	97	99	101
Achieved read:	88	70	90	81	91	95	93

Calculate the F value and other statistics from the achieved math and reading scores.

Using the above achieved math scores, predict the reading scores:

Achieved math:	70	75	80	80	97	99	101
Achieved read:	79.8	81.9	84.2	84.2	91.7	92.6	93.5

Use the achieved math scores of 25, 50, and 76 to predict reading scores:

Achieved math:	25	50	76
Predicted read:	59.8374	70.9111	82.427

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "PREDICT"	Linear Regression Prediction	
1	Enter raw data	Enter raw data (Y/N) ?	Y [RTN]
2		RTN to end data, D to delete	
		Point #1 Enter X , Y ?	70,88 [RTN]
		Point #2 Enter X , Y ?	75,70 [RTN]
		Point #3 Enter X , Y ?	80,90 [RTN]
	Error	Point #4 Enter X , Y ?	80,811 [RTN]
	Call deletion function	Point #5 Enter X , Y ?	D,D [RTN]
		Point #4 Deleted: X=80 Y=811	
	Enter correct values	Point #4 Enter X , Y ?	80,81 [RTN]
		Point #5 Enter X , Y ?	97,91 [RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Point #6 Enter X , Y ?	99,95 [RTN]
		Point #7 Enter X , Y ?	101,93 [RTN]
	End data input	Point #8 Enter X , Y ?	[RTN]
3	$\bar{x}$ and s	X Mean = 86 s =11.7352	[RTN]
	$\bar{y}$ and s	Y Mean = 86.8571 s = 8.0255	[RTN]
	Correlation coefficient	Correlation xy = .6477	[RTN]
	f-value	F = 3.6134	[RTN]
	Standard error	St. error = 6.1146	[RTN]
	Slope and intercept of line	Byx (Slope) = .4429	[RTN]
		Ayx (Intercept) = 48.7638	[RTN]
4	Predictions:	Prediction. Type E to exit	
		Enter the known X?	70 [RTN]
		X = 70 Pred.Y = 79.77	[RTN]
		Enter the known X?	75 [RTN]
	NOTE that the range of the prediction is printed, but not shown here. (Range = prediction ± standard error).		
		X = 75 Pred.Y = 81.9847	[RTN]
		Enter the known X?	80 [RTN]
		X = 80 Pred.Y = 84.1995	[RTN]
		Enter the known X?	97 [RTN]
		X = 97 Pred.Y = 91.7295	[RTN]
		Enter the known X?	99 [RTN]
		X = 99 Pred.Y = 92.6154	[RTN]
		Enter the known X?	101 [RTN]
		X = 101 Pred.Y = 93.5013	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
		Enter the known X?	25 [RTN]
		X = 25 Pred.Y = 59.8374	[RTN]
		Enter the known X?	50 [RTN]
		X = 50 Pred.Y = 70.9111	[RTN]
		Enter the known X?	76 [RTN]
		X = 76 Pred.Y = 82.4277	[RTN]
5	Exit	Enter the known X?	E [RTN]

SAMPLE PROBLEM

To continue the study in sample problem 1 after the machine has been cleared without re-entering the raw data use $\overline{x}$=86, $\overline{y}$=86.8571, s_x=11.7352, s_y=8.0255, r_{xy}=.6477, n=7.

Predict the reading scores from the math scores of 27 and 150. NOTE: It would be possible to predict the reading scores by reversing the values of $\overline{x}$, $\overline{y}$, s_x, and s_y.

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run	Linear Regression Prediction	
1	Enter statistics:	Input raw data (y/n)?	N [RTN]
2		Enter Correlation Coeff.?	.6477 [RTN]
		Enter the mean of X?	86 [RTN]
		Enter the SD of X?	11.7352 [RTN]
		Enter the mean of Y?	86.8571 [RTN]
		Enter the SD of Y?	8.0255 [RTN]
		Enter the sample size (n)?	7 [RTN]
3	F value	F=3.6135	[RTN]
	Standard Error	St. error=6.1146	[RTN]
	Slope and intercept of line	Byx (Slope)=.443	[RTN]
		Ayx (Intercept)=48.562	[RTN]
4	Prediction	Enter the known X?	27 [RTN]
		X=27 Pred.Y=60.74	[RTN]
	Confidence interval	Range Y [54.6013-66.8305]	[RTN]
	Prediction	Enter the known X?	150 [RTN]
		X=150 Pred.Y=155.1989	[RTN]
	Confidence interval	Range y[109.0843-121.3134]	[RTN]
5	End	Enter the known X?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "PREDICT"	Linear Regression Prediction	
1	If entering raw data goto 3	Enter raw data (Y/N)?	Y or N [RTN]
2	Enter statistics	Enter Correlation Coeff.?	Rxy [RTN]
		Enter the mean of X?	$\overline{x}$ [RTN]
		Enter the S.D. of X?	s_x [RTN]
		Enter the mean of Y?	$\overline{y}$ [RTN]
		Enter the S.D. of Y?	s_y [RTN]
		Enter the sample size (N)?	N [RTN]
	Goto step 5		
3	Enter raw data	RTN to end data, D,D to delete	
	Repeat step 3 for all data. If the last pair was wrong, enter	Point #n Enter X , Y ?	X,Y [RTN]
	D,D and then re-enter the pair. [RTN] when all N pairs are in.	Point #n+1 Enter X , Y ?	[RTN]
4	View statistics	X mean = $\overline{x}$ s=s_x	[RTN]
	Use [BACK] to view the pre-	Y Mean = $\overline{y}$ s=s_y	[RTN]/[BACK]
	vious answer	Correlation XY = Rxy	[RTN]/[BACK]
5	F value	F = value	[RTN]/[BACK]
	Standard Error	St. Error = value	[RTN]/[BACK]
	Slope and intercept of line	Byx (Slope) = value	[RTN]/[BACK]
		Ayx (Intercept) = value	[RTN]/[BACK]
6		Prediction	
	Predict Y from X	Enter the known X?	x [RTN]
	Predicted Y	X = value Pred.Y = value	[RTN]
	Confidence interval	Range Y[value - value]	[RTN]
7	Enter 'E' to end, else goto 6	Enter the known X?	E [RTN]

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
X( )	X-value storage	X$	X-value input string
Y( )	Y-value storage	Y$	Y-value input string
X1	Sum of X Σx	Y1	Sum of Y Σy
X2	Sum of X squared Σx^2	Y2	Sum of Y squared Σy^2
C	Cross product Σxy	N	Number of data
E	Standard error	K$	General input string
M1	Mean of X	M2	Mean of Y
S1	Standard deviation of X	S2	Standard deviation of Y
F	F-value	R	Correlation coefficient
E1	Min. confidence interval	E2	Max. confidence interval
U	Delay value: 1=[BACK] key	A	Axy: Y intercept
B	Bxy: Slope of line		

NOTES AND REFERENCES

A maximum of 100 data points may be entered as raw data. To change this limit, change the dimension statement in line 60.

The standard deviation presented is the population. In cases where a sample of the population is used and the number of data points is less than 30, change line 530 as follows:

```
530 M1=X1/N @ M2=Y1/N @ S1=SQR((N*X2-X1^2)/(N*(N-1))) @ S2=SQR((N*Y2-Y1^2)/(N*(N-1)))
```

References: Cohen, J. & P., APPLIED MULTIPLE REGRESSION/CORRELATION ANALYSIS FOR THE BEHAVIORAL SCIENCES (Lawrence Erlbaum Associates, New Jersey, 1975) pages 39-43, 110.

PROGRAM LISTING

```
 10 ! PREDICT - Linear
 20 ! prediction
 30 ! REV 11/01/82
 40 !
 50 DISP "Linear Regression Prediction"
 60 DIM X(100),Y(100),X$[100],Y$[100]
 70 X1,X2,Y1,Y2,C,N=0
 80 DEF FNI(X) = INT(X*10000+.5)/10000    -Function to set output
                                           precision
 90 !
100 ! Delay Routine
110 !
120 DEF FND                               -Wait until a key is pressed,
                                           return 1 if BACK
130 BEEP 440
140 K$=KEY$ @ IF POS(CHR$(13)&CHR$(8)&"
    E",UPRC$(K$))=0 THEN 140
150 IF UPRC$(K$)="E" THEN 820
160 FND=K$=CHR$(8)
170 END DEF
180 !
190 ! Error correction
200 !
210 DEF FNE                               -Error correction routine -
                                           delete last input pair
220 FNE=1
230 IF N<1 THEN BEEP 1760 @ DISP "Must
    have data to delete " @ GOTO 280
240 DISP "Point #";N;"Deleted: X=";X(N)
    ;"Y=";Y(N) @ WAIT 1
250 X1=X1-X(N) @ Y1=Y1-Y(N)
260 X2=X2-X(N)^2 @ Y2=Y2-Y(N)^2 @ C=C-X
    (N)*Y(N)
270 N=N-1
280 END DEF
290 INPUT "Enter raw data (Y/N) ?";X$     -Determine data type - raw or
                                           direct?
300 IF UPRC$(X$[1,1])="Y" THEN 420
310 INPUT "Enter Correlation Coeff. ?";   -Enter values of Mean, Sd. and
    R                                      r directly
320 IF ABS(R)>1 THEN BEEP @ GOTO 310
330 INPUT "Enter the mean of X ?";M1
340 INPUT "Enter the S.D. of X ?";S1
350 INPUT "Enter the mean of Y ?";M2
360 INPUT "Enter the S.D. of Y ?";S2
370 INPUT "Enter the sample size (N) ?"
    ;N
380 GOTO 620
390 !
400 ! Enter raw data
410 !
420 DISP "RTN to end data  ";CHR$(196);  -Raw data entry- compute Mean,
    " to delete"; @ WAIT 1 @ ON ERROR G    Sd, and r.
    OTO 460
```

PROGRAM LISTING

```
430 DISP "Point #";N+1;" Enter X , Y ";
440 INPUT X$,Y$ @ IF X$="" THEN 530
450 IF POS("DdEe",X$[1,1]) THEN U=FNE @
     GOTO 430 ELSE GOTO 470
460 IF ERRN=81 THEN OFF ERROR @ GOTO 53
    0 ELSE GOTO 430
470 N=N+1 @ X(N)=VAL(X$) @ Y(N)=VAL(Y$)    -Increment counters
480 X1=X1+X(N) @ X2=X2+X(N)^2 @ Y1=Y1+Y
    (N) @ Y2=Y2+Y(N)^2
490 C=C+X(N)*Y(N) @ GOTO 430
500 !
510 ! Compute mean,s
520 !
530 M1=X1/N @ M2=Y1/N @ S1=SQR(X2/N-(X1    -Compute Mean and Sd of X and Y
    /N)^2) @ S2=SQR(Y2/N-(Y1/N)^2)
540 R=(C-X1*Y1/N)/SQR((X2-X1^2/N)*(Y2-Y    -Compute correlation between X
    1^2/N))                                  and Y
550 PRINT "X Mean=";FNI(M1);"s=";FNI(S1    -Print mean and Sd.
    ) @ U=FND
560 PRINT "Y Mean=";FNI(M2);"s=";FNI(S2
    )
570 U=FND @ IF U THEN 550
580 PRINT "Correlation xy =";FNI(R) @ U
    =FND @ IF U THEN 560
590 !
600 ! F,St. error
610 !
620 F=R^2/((1-R^2)/(N-2))                  -Compute F
630 E=S2*SQR(1-R^2)                        -Compute standard error
640 B=R*(S2/S1) @ A=M2-B*M1                -Compute slope and Y-intercept
650 PRINT "F=";FNI(F) @ U=FND @ IF U TH
    EN 580
660 PRINT "St. Error =";FNI(E) @ U=FND
    @ IF U THEN 650
670 PRINT "Byx (Slope) =";FNI(B) @ U=FN
    D @ IF U THEN 660
680 PRINT "Ayx (Intercept) =";FNI(A) @
    U=FND @ IF U THEN 670
690 !
700 ! Prediction routine
710 !
720 DISP "Prediction. Type ";CHR$(197);
    " to exit" @ WAIT 1
730 ON ERROR BEEP 440 @ DISP "Please en    -Error trapping routine
    ter Numeric or 'E'" @ WAIT 1 @ GOTO
     740
740 INPUT "Enter the known X ?";X$
750 IF UPRC$(X$[1,1])="E" THEN 820
760 X=VAL(X$)
770 Y=R*S2*((X-M1)/S1)+M2                  -Predict Y from X
780 E1=Y-E @ E2=Y+E
790 PRINT "X=";X;"Pred. Y=";FNI(Y) @ U=    -Display confidance interval
    FND
800 PRINT "Range Y[";STR$(FNI(E1));" -
    ";STR$(FNI(E2));"]" @ U=FND @ IF U
    THEN 790
```

PROGRAM LISTING

```
810 GOTO 740
820 STOP
```

PROGRAM DESCRIPTION

PERMUTATIONS AND COMBINATIONS

This program will calculate the number of permutations and combinations for a given number of objects, n, divided into groups of a given size r. A permutation is the number of items that can be fit into a specified order, and a combination is the number of items that may be fit without any specific order.

The formula for the number of permutations of n objects in groups of size r is:

$$nPr = \frac{n!}{(n-r)!}$$

(! = Factorial = 1*2*3*.....*n)

The number of combinations of n objects in groups of size r is:

$$nCr = \frac{nPr}{r!}$$

SAMPLE PROBLEM

1) Suppose a restaurant has 10 chairs at a table. How many ways can 10 guests be seated at the table? (3628800)

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
A	Run "PERMUT"	Permutations and Combinations	
1	Enter the number of guests	Enter the number of objects?	10 [RTN]
2	Enter the number of chairs	Enter the group size?	10 [RTN]
3	Answer: # of Permutations	10P10 = 3628800 Permutations	[RTN]
	# of Combinations	10C10 = 1 Combinations	[RTN]
4	Re-run the program	Run again, View again or End? R	[RTN]

SAMPLE PROBLEM

2) A city has a council with 3 vacant seats and 40 candidates. How many ways could the three seats be filled? (9880)

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
B	Run "PERMUT"	Permutations and Combinations	
1	Enter the number of candidates	Enter the number of objects?	40 [RTN]
2	Enter the number of seats	Enter the group size?	3 [RTN]
3	# of Permutations	40P3 = 59280 Permutations	[RTN]
	Answer: # of Combinations	10C3 = 9880 Combinations	[RTN]
4	Stop	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run "PERMUT"	Permutations and Combinations	
1	Enter n	Enter the number of objects?	n [RTN]
2	Enter c	Enter the group size?	c [RTN]
3	Read output	nPr = P Permutations	[RTN]
		nCr = C Combinations	[RTN]
4	Review routine	Run again, View again, or End?	R, V, or E [RTN]
	R = re-run program		
	V = re-view results		
	E = end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
N	Number of objects	P	Permutations
R	Group size	K$	Option entry
C	Combinations	I	Counter

NOTES AND REFERENCES

Between lines 110 and 150, this program has a routine to calculate the factorial of any number ≠>0. To use this routine in other programs, enter lines 110-150 of this program, and use the routine in the following manner:

```
(Sample)            1050 input "X= ?";x
                    1060 disp "X Factorial = ";FNF(X)
(Sample formula)    1070 y=fnf(x)/x
```

References: Gehring, Robert E., BASIC BEHAVIORAL STATISTICS (Houghten-mifflen, Boston, 1978) page 418-422.

PROGRAM LISTING

```
 10 ! PERMUT - Factorial
 20 ! Permutations and
 30 ! Combinations
 40 ! REV 11/01/82
 50 !
 60 DISP " Permutations and Combination
    s"
 70 DIM K$[20]
 80 !
 90 ! Factorial
100 !
110 DEF FNF(N)                          -Routine to compute the
                                         factorial of a number
120 V=1 @ FOR I=2 TO N
130 V=V*I
140 NEXT I @ FNF=V
150 END DEF
160 !
170 ! Error display
180 !
190 ON ERROR BEEP 880 @ DISP "Overflow!  -Error trapping routine for
    " @ GOTO 230                         overflow
200 !
210 ! Input dialogue
220 !
230 INPUT "Enter the number of objects
    ?";N
240 INPUT "Enter the group size ?";R
250 !
260 ! Calculation
270 !
280 N=INT(ABS(N)) @ R=INT(ABS(R))
290 IF N<R THEN BEEP @ GOTO 230
300 P=FNF(N)/FNF(N-R)                   -Calculate the number of
                                         permutations
310 C=P/FNF(R)                          -Calculate the number of
                                         combinations
320 C=INT(C) @ P=INT(P)
330 !
340 ! Output
350 !
360 PRINT STR$(N);"P";STR$(R);" =";P;"P  -Print the number of
    ermutations"                         permutations
370 BEEP 440
380 IF KEY$="" THEN 380
390 PRINT STR$(N);"C";STR$(R);" =";C;"C  -Calculate the number of
    ombinations"                         combinations
400 BEEP 440
410 IF KEY$="" THEN 410
420 !
430 ! Recycle
440 !
450 DISP CHR$(210);"un again, ";CHR$(21  -Recycle routine
    4);"iew again, or ";CHR$(197);"nd";
```

PROGRAM LISTING

```
460 INPUT K$ @ K$=UPRC$(K$&" ")
470 ON POS("RVE",K$[1,1])+1 GOTO 450,23
    0,360,480
480 STOP
```

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

STATISTICS

BASIC ONE VARIABLE STATISTICS
COEFFICIENT OF CORRELATION
PROBABILITY OF NORMAL, F, T, AND CHI-SQUARE DISTRIBUTIONS
DEPENDENT (PAIRED) T-TEST
T-TEST FOR 2 UNEQUAL SIZED SAMPLES
CHI-SQUARE TEST
ONE-WAY ANALYSIS OF VARIANCE
SIMPLE LINEAR REGRESSION
PERMUTATIONS AND COMBINATIONS

ALL HP-75 SOLUTIONS BOOKS ARE AVAILABLE RECORDED ON MINI-DATA CASSETTES FROM EITHER A HEWLETT-PACKARD DEALER OR THE HP USERS' LIBRARY.

00075-90067 Printed in U.S.A.

Reorder No.
00075-13011